****
****	Add Health replication.do
****
****
****	The script below replicates the Add Health portion of the paper:
****	"Depression and Political Participation" forthcoming in Social Science Quaterly
****
****

**
** Read in data-- need to set path file name

*cd "PATH FILE NAME HERE"
*use "DATA FILE NAME HERE"


**
** Determining sample size based on final model specification
svyset [weight=gswgt3_2]
svy: logistic civ_part_dum mhealth_scale_ad mhealth_scale3 female race_black race_other calcage3 educ_years parent_income parent_hs parent_college pa_org_dum gen_health_scale3
predict y_part
gen y_samp = y_part
replace y_samp = . if civ_part_dum == . | gswgt3_2 == .


**
** Descriptive statistics

* mean and sd of dependent variables
sum civ_vote civ_part_dum if y_samp != . [aw=gswgt3_2]

* correlation of wave 1 and wave 2 depressed mood scales
cor mhealth_sum1_9 mhealth_sum2_9 if y_samp != . [aw=gswgt3_2]

* mean and sd of transformed depressed mood scale
sum mhealth_scale3 mhealth_scale_ad if y_samp != . [aw=gswgt3_2]

* alpha score of items for adolescent depressed mood scale
alpha mhealth1 mhealth3 mhealth4 mhealth5 mhealth6 mhealth7 mhealth15 mhealth16 mhealth17 mhealth2_1 mhealth2_3 mhealth2_4 mhealth2_5 mhealth2_6 mhealth2_7 mhealth2_15 mhealth2_16 mhealth2_17, item

* alpha score of items for adult depressed mood scale
alpha mhealth3_1 mhealth3_3 mhealth3_4 mhealth3_5 mhealth3_6 mhealth3_7 mhealth3_15 mhealth3_16 mhealth3_17 if y_samp != ., item

* mean and sd of control variables in multivariate analysis		 
sum female race_black race_other calcage3 educ_years parent_inc10 parent_hs parent_college pa_org_dum gen_health_scale3 if y_samp != . [aw=gswgt3_2]


**
** Results of multivariate analysis
			 
svyset [weight=gswgt3_2]

* voter turnout
svy: logistic civ_vote mhealth_scale_ad mhealth_scale3 female race_black race_other calcage3 educ_years gen_health_scale3 parent_inc10 parent_hs parent_college pa_org_dum
prchange, rest(median)

* non-voting participation
svy: logistic civ_part_dum mhealth_scale_ad mhealth_scale3 female race_black race_other calcage3 educ_years gen_health_scale3 parent_inc10 parent_hs parent_college pa_org_dum
prchange, rest(median)


**
** Endnotes

* concern with multicollinearity between adolsecent and young adult depressed mood 
cor mhealth_scale_ad mhealth_scale3 if y_samp != .
tab mhealth_scale_ad if y_samp!=. & mhealth_scale3 <= 0.04
tab mhealth_scale_ad if y_samp!=. 
svyset [weight=gswgt3_2]
svy: logistic civ_vote mhealth_scale_ad female race_black race_other calcage3 educ_years gen_health_scale3 parent_inc10 parent_hs parent_college pa_org_dum if mhealth_scale3 <= .04
prchange, rest(median)
svy: logistic civ_part_dum mhealth_scale_ad female race_black race_other calcage3 educ_years gen_health_scale3 parent_inc10 parent_hs parent_college pa_org_dum if mhealth_scale3 <= .04
prchange, rest(median)

